



Techniques of Water-Resources Investigations
of the United States Geological Survey

Chapter A1

**A MODULAR THREE-DIMENSIONAL
FINITE-DIFFERENCE GROUND-WATER
FLOW MODEL**

By Michael G. McDonald and
Arlen W. Harbaugh

This chapter supersedes U.S. Geological
Survey Open-File Report 83-875

Book 6

MODELING TECHNIQUES

Narrative for Module SSOR1B

Module SSOR1B uses Gaussian elimination to solve a matrix equation of the form $[A]\{X\}=\{B\}$. The matrix $[A]$ is symmetric and banded with rank "N." It is stored in the compressed format shown in figure 55 and uses a single subscript. The vector $\{X\}$, as it is calculated, is stored in the space reserved for vector $\{B\}$.

The indices used in the module flow chart are those for standard matrix organization. The indices actually used in the program are based on the compressed format and a single index. Module SSOR1B performs its functions in the following order:

1. Work through the first N-1 rows using each one, in sequence, as the pivot row (row I).

2. Calculate the inverse of the main diagonal element--- $a_{I,I}$. (The index (ID) points to $a_{I,I}$ which is the first element of column I in the compressed matrix $[A]$.)

3. Modify each of the rows after row I so that the terms corresponding to the pivot term are eliminated. Since the coefficient matrix is banded, there are only MBW-1 equations (where MBW is the maximum half-bandwidth plus one) where the term to be eliminated is not already equal to zero. The rows are indexed by "L." (The equation corresponding to row "L" is referred to as equation "L.")

4. Calculate the coefficient C which when multiplied by the pivot equation and subtracted from equation L will eliminate a term in equation L.

(The index IB points to a coefficient in the pivot equation which, because of symmetry, is equal to the coefficient to be eliminated.)

5. Calculate the new coefficients in equation L for each of the terms to the right of the coefficient that is being eliminated. Because the matrix is banded, there are only MBW-1 nonzero terms to the right of the pivot. Therefore, at most, MBW-1 coefficients have to be calculated.

6. Subtract C times a coefficient in the pivot equation from the corresponding coefficient in equation L.

7. Subtract C times the right side of the pivot equation from the right side of equation L. (The index LB points to the coefficient in equation L which must be calculated.)

8. Solve equation N for $X(N)$ putting the result in $B(N)$.

9. Work backward from equation N-1 solving each equation (equation L) for $X(L)$.

10. Set the accumulator "SUM" equal to zero.


11. Multiply the coefficient to the right of the main diagonal (in equation L) by the corresponding value of X and add it to the sum.

12. Calculate the value of $X(L)$ and store it in $B(L)$.

13. RETURN.

Flow Chart for Module SSOR1B

MBW is the maximum half bandwidth of the coefficient matrix plus one.



```

SUBROUTINE SSOR1B(A,B,N,NA,MBW)
C
C
C-----VERSION 1359 31MAR1983 SSOR1B
C *****
C SOLVE A SYMMETRIC SET OF EQUATIONS
C   A IS COEFFICIENT MATRIX IN COMPRESSED FORM
C   B IS RIGHT HAND SIDE AND IS REPLACED BY SOLUTION
C   N IS NUMBER OF EQUATIONS TO BE SOLVED
C   MBW IS BANDWIDTH OF A
C   NA IS ONE-DIMENSION SIZE OF A
C *****
C
C SPECIFICATIONS:
C -----
C DIMENSION A(NA),B(N)
C -----
C
C   NM1=N-1
C   MBW1=MBW-1
C   ID=1-MBW
C
C1-----SEQUENTIALLY USE EACH OF THE FIRST N-1 ROWS AS
C1-----THE PIVOT ROW.
C   DO 20 I=1,NM1
C
C2-----CALCULATE THE INVERSE OF THE PIVOT.
C   ID=ID+MBW
C   C1=1./A(ID)
C   LD=ID
C   L=I
C
C3-----FOR EACH ROW AFTER THE PIVOT ROW (THE TARGET ROW)
C3-----ELIMINATE THE COLUMN CORRESPONDING TO THE PIVOT.
C   DO 15 J=1,MBW1
C   L=L+1
C   IF(L.GT.N) GO TO 20
C   IB=ID+J
C
C4-----CALCULATE THE FACTOR NEEDED TO ELIMINATE A TERM IN THE
C4-----TARGET ROW.
C   C=A(IB)*C1
C   LD=LD+MBW

```

```

      LB=LD-1
C
C5-----MODIFY THE REST OF THE TERMS IN THE TARGET ROW.
      DO 10 K=J,MBW1
C
C6-----SUBTRACT THE FACTOR TIMES A TERM IN THE PIVOT ROW
C6-----FROM THE CORRESPONDING COLUMN IN THE TARGET ROW.
      LB=LB+1
      A(LB)=A(LB)-C*A(ID+K)
      10 CONTINUE
C
C7-----MODIFY THE RIGHT SIDE OF THE EQUATION CORRESPONDING
C7-----TO THE TARGET ROW.
      B(I+J)=B(I+J)-C*B(I)
      15 CONTINUE
      20 CONTINUE
      ID=ID+MBW
C
C8-----SOLVE THE LAST EQUATION.
      B(N)=B(N)/A(ID)
C
C9-----WORKING BACKWARDS SOLVE THE REST OF THE EQUATIONS.
      DO 70 I=1,NM1
      ID=ID-MBW
C
C10-----CLEAR THE ACCUMULATOR SUM.
      SUM=0.0
      L=N-I
      MBWIM=MINO(MBW1,I)
C
C11-----ADD THE KNOWN TERMS IN EQUATION L TO SUM.
      DO 60 J=1,MBWIM
      SUM=SUM+A(ID+J)*B(L+J)
      60 CONTINUE
C
C12-----SOLVE FOR THE ONE UNKNOWN IN EQUATION L.
      B(L)=(B(L)-SUM)/A(ID)
      70 CONTINUE
C
C13-----RETURN
      RETURN
      END

```

List of Variables for Module SSOR1B

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
A	Package	DIMENSION (MBW,NSLICE), Compressed coefficient matrix for a slice.
B	Package	DIMENSION (N), Right-hand-side vector.
C	Module	Factor needed to eliminate a term in the target row.
C1	Module	Inverse of pivot.
I	Module	Index for rows in the SSOR matrix (not the grid).
IB	Module	Index for elements to the right of the pivot.
ID	Module	Index of pivots.
J	Module	Index for columns.
K	Module	Index for columns.
L	Module	Index for equations.
LB	Module	Index for elements in the target row to the right of the main diagonal.
LD	Module	Index of the main diagonal elements in the target rows.
MBW	Package	Maximum bandwidth of the coefficient matrix +1.
MBW1	Module	Maximum bandwidth of the coefficient matrix.
MBW1M	Module	Maximum possible number of the nonzero elements to the right of the main diagonal.
N	Package	Number of equations to be solved.
NA	Package	One-dimension size of compressed matrix "A."
NM1	Module	N-1.
SUM	Module	In back substitution--in equation L, sum of terms to the right of the main diagonal term (L,L).

CHAPTER 14
UTILITY MODULES

Utility modules are those submodules which perform general tasks common to several different packages. The name of a utility module always consists of a "U" followed by a five-character mnemonic. There are eight utility modules:

UBUDSV	Writes an unformatted record consisting of an array with one real number for each cell in the grid.
ULASAV	Writes an unformatted record consisting of an array with one real number for each cell in a layer.
ULAPRS and ULAPRW	Prints one two-dimensional array which contains one real number for each cell in a layer. ULAPRS prints, in strip form, the first N columns (where N is the number of values that can fit on one print line) of each row and then the next N columns, etc., until all columns of each row are printed (fig. 56). ULAPRW prints, in wrap form, all of row 1, all of row 2, and all of row 3, etc. The format for printing arrays is shown in table 2.
UCOLNO	Prints column numbers at the top of each page of data printed by ULAPRS and ULAPRW.
U2DREL	Reads a two-dimensional array of real numbers.
U2DINT	Reads a two-dimensional array of integers.
U1DREL	Reads a one-dimensional array of real numbers.

	1	2	3	4	5	6	7	8	9	10
	11	12	13	14	15	16	17			
1	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
2	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
3	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
4	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
5	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
6	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
7	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79

WRAP FORM

	1	2	3	4	5	6	7	8	9	10
1	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
2	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
3	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
4	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
5	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
6	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
7	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79

	11	12	13	14	15	16	17
1	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
2	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
3	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
4	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
5	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
6	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79
7	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79	1325.79

STRIP FORM

Figure 56.—Illustration of wrap and strip forms of printed output for a layer containing 7 rows and 17 columns.

Table 2.--Print-format codes for utility modules
ULAPRS and ULAPRW.

<u>IPRN</u>	<u>FORMAT</u>
1	11G10.3
2	9G13.6
3	15F7.1
4	15F7.2
5	15F7.3
6	15F7.4
7	20F5.0
8	20F5.1
9	20F5.2
10	20F5.3
11	20F5.4
12	10G11.4

Input Instructions For Array Reading Utility Modules

The real two-dimensional array reader (U2DREL), the integer two-dimensional array reader (U2DINT), and the real one-dimensional array reader (U1DREL) read one array-control record and, optionally, a data array in a format specified on the array-control record. The control record is read from the input unit number specified for the major option that is requesting the array. For example, the Recharge Package uses U2DREL to read the RECH array. The input unit for the recharge option is contained in IUNIT (8), and accordingly, the RECH array control record is read on this input unit.

FOR REAL ARRAY READER (U2DREL or U1DREL)

Data:	LOCAT	CNSTNT	FMTIN	IPRN
Format:	I10	F10.0	5A4	I10

FOR INTEGER ARRAY READER (U2DINT)

Data:	LOCAT	ICONST	FMTIN	IPRN
Format:	I10	I10	5A4	I10

Explanation of Fields Used in Input Instructions

LOCAT--indicates the location of the data which will be put in the array.

If LOCAT < 0, the sign is reversed to give the unit number from which an unformatted record will be read.

If LOCAT = 0, every element in the array will be set equal to the value CNSTNT/ICONST.

If LOCAT > 0, it is the unit number from which data values will be read in the format specified in the third field of the array-control record (FMTIN).

CNSTNT/ICONST--is a constant. Its use depends on the value of LOCAT.

If LOCAT = 0, every element in the array is set equal to CNSTNT/ICONST.

If LOCAT ≠ 0, and if CNSTNT/ICONST ≠ 0, every element in the array is multiplied by CNSTNT/ICONST.

FMTIN--is the format of records containing the array values. It is used only if the first field in the array-control record (LOCAT) contains a positive number. The format must be enclosed in parentheses; for example, (15F5.0) for real data and (15I5) for integer data.

IPRN--is a flag indicating that the array being read should be printed and a code for indicating the format that should be used. It is used only if LOCAT is not equal to zero. The format codes are different for each of the three modules. IPRN is set to zero when the specified value exceeds those defined in the chart below. If IPRN is less than zero, the array will not be printed.

<u>IPRN</u>	<u>U2DREL</u>	<u>U2DINT</u>	<u>U1DREL</u>
0	10G11.4	10I11	10G12.5
1	11G10.3	60I1	
2	9G13.6	40I2	
3	15F7.1	30I3	
4	15F7.2	25I4	
5	15F7.3	20I5	
6	15F7.4		
7	20F5.0		
8	20F5.1		
9	20F5.2		
10	20F5.3		
11	20F5.4		
12	10G11.4		

Narrative for Module UBUDSV

Utility module UBUDSV writes an unformatted record consisting of an array dimensioned (NCOL, NROW, NLAY). The record containing the array is preceded by an unformatted record containing identifying information. The identifying information consists of:

KSTP	current time step	integer	1 word
KPER	current stress period	integer	1 word
TEXT	label	character string	4 words
NCOL	number of columns	integer	1 word
NROW	number of rows	integer	1 word
NLAY	number of layers	integer	1 word

Documentation of this module consists only of comments in the program and a list of variables.

```

SUBROUTINE UBUDSV (KSTP, KPER, TEXT, IBDCHN, BUFF, NCOL, NROW, NLAY, IOUT)
C
C
C-----VERSION 1637 12MAY1987 UBUDSV
C *****
C RECORD CELL-BY-CELL FLOW TERMS FOR ONE COMPONENT OF FLOW.
C *****
C
C SPECIFICATIONS:
C -----
C CHARACTER*4 TEXT
C DIMENSION TEXT(4),BUFF(NCOL,NROW,NLAY)
C -----
C
C1-----WRITE AN UNFORMATTED RECORD CONTAINING IDENTIFYING
C1-----INFORMATION.
C WRITE(IOUT,1) TEXT,IBDCHN,KSTP,KPER
C 1 FORMAT(1X,'"',4A4,'" BUDGET VALUES WILL BE SAVED ON UNIT',I3,
C 1 ' AT END OF TIME STEP',I3,' STRESS PERIOD',I3)
C
C WRITE(IBDCHN) KSTP,KPER,TEXT,NCOL,NROW,NLAY
C
C2-----WRITE AN UNFORMATTED RECORD CONTAINING VALUES FOR
C2-----EACH CELL IN THE GRID. THE ARRAY IS DIMENSIONED
C2----- (NCOL,NROW,NLAY)
C WRITE(IBDCHN) BUFF
C
C3-----RETURN
C RETURN
C END

```

List of Variables for Module UBUDSV

<u>Variation</u>	<u>Range</u>	<u>Definition</u>
BUFF	Global	DIMENSION (NCOL,NROW,NLAY), Buffer used to accumulate information before printing or recording it.
IBDCHN	Module	Unit number on which the array will be recorded.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
TEXT	Module	Label to be printed or recorded with the array data.

Narrative for Module ULASAV

Utility module ULASAV writes an unformatted record consisting of an array dimensioned (NCOL, NROW). The record containing the array is preceded by an unformatted record containing identifying information. The identifying information consists of:

KSTP	current time step	integer	1 word
KPER	current stress period	integer	1 word
PERTIM	elapsed time in the current stress period	real	1 word
TOTIM	elapsed time in the simulation	real	1 word
TEXT	label	character string	4 words
NCOL	number of columns	integer	1 word
NROW	number of rows	integer	1 word
ILAY	layer number	integer	1 word


```

SUBROUTINE ULASAV(BUF, TEXT, KSTP, KPER, PERTIM, TOTIM, NCOL,
1 NROW, ILAY, ICHN)
C
C-----VERSION 1642 12MAY1987 ULASAV
C *****
C SAVE 1 LAYER ARRAY ON DISK
C *****
C
C SPECIFICATIONS:
C -----
C CHARACTER*4 TEXT
C DIMENSION BUF(NCOL, NROW), TEXT(4)
C -----
C
C1-----WRITE AN UNFORMATTED RECORD CONTAINING IDENTIFYING
C1-----INFORMATION.
C WRITE(ICHN) KSTP, KPER, PERTIM, TOTIM, TEXT, NCOL, NROW, ILAY
C
C2-----WRITE AN UNFORMATTED RECORD CONTAINING ARRAY VALUES
C2-----THE ARRAY IS DIMENSIONED (NCOL, NROW)
C WRITE(ICHN) ((BUF(IC, IR), IC=1, NCOL), IR=1, NROW)
C
C3-----RETURN
C RETURN
C END

```

List of Variables for Module ULASAV

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
BUF	Module	Buffer containing data to be printed or recorded.
IC	Module	Index for columns.
ICHN	Module	Unit number on which the array is to be recorded.
ILAY	Module	Layer number.
IR	Module	Index for rows.
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
NCOL	Global	Number of columns in the grid.
NROW	Global	Number of rows in the grid.
PERTIM	Package	Elapsed time during the current stress period.
TEXT	Module	Label to be printed or recorded with the array data.
TOTIM	Package	Elapsed time in the simulation.


Narrative for Module ULAPRS

Module ULAPRS prints a two-dimensional array in strip form (fig. 56) using one of twelve FORTRAN formats. Module ULAPRS performs its tasks in the following order:

1. Get the format code (IP). If it is less than 1 or greater than 12, set it equal to 12 (the default).
2. Use the format code (IP) to determine the number of values (NCAP) to be printed on one line.
3. Calculate the number of spaces used for each value (NCPF) and the number of strips (NSTRIP). Initialize the fields to store the first column (J1) and the last column (J2) for each strip.
4. Loop through the strips (DO STEPS 5-8).
5. Calculate the first (J1) and last (J2) column for this strip.
6. Print a title on each strip.
7. Call module UCOLNO to print the column numbers above each strip.
8. Loop through the rows printing columns J1 through J2 using the appropriate format (IP).
9. RETURN.

Flow Chart for Module ULAPRS

IPRN is a code indicating the format to be used in printing array values. If it is not between 1 and 12, it is set equal to 12.



```

SUBROUTINE ULAPRS(BUF,TEXT,KSTP,KPER,NCOL,NROW,ILAY,IPRN,IOUT)
C
C
C-----VERSION 1640 12MAY1987 ULAPRS
C *****
C PRINT A 1 LAYER ARRAY IN STRIPS
C *****
C
C SPECIFICATIONS:
C -----
C CHARACTER*4 TEXT
C DIMENSION BUF(NCOL,NROW),TEXT(4)
C -----
C
C1-----MAKE SURE THE FORMAT CODE (IP OR IPRN) IS BETWEEN 1
C1-----AND 12.
C IP=IPRN
C IF(IP.LT.1 .OR. IP.GT.12) IP=12
C
C2-----DETERMINE THE NUMBER OF VALUES (NCAP) PRINTED ON ONE LINE.
C IF(IP.EQ.1) NCAP=11
C IF(IP.EQ.2) NCAP=9
C IF(IP.GT.2 .AND. IP.LT.7) NCAP=15
C IF(IP.GT.6 .AND. IP.LT.12) NCAP=20
C IF(IP.EQ.12) NCAP=10
C
C3-----CALCULATE THE NUMBER OF STRIPS (NSTRIP).
C NCPF=129/NCAP
C ISP=0
C IF(NCAP.GT.12) ISP=3
C NSTRIP=(NCOL-1)/NCAP + 1
C J1=1-NCAP
C J2=0
C
C4-----LOOP THROUGH THE STRIPS.
C DO 2000 N=1,NSTRIP
C
C5-----CALCULATE THE FIRST(J1) & THE LAST(J2) COLUMNS FOR THIS STRIP
C J1=J1+NCAP
C J2=J2+NCAP
C IF(J2.GT.NCOL) J2=NCOL
C
C6-----PRINT TITLE ON EACH STRIP
C WRITE(IOUT,1) TEXT,ILAY,KSTP,KPER
C 1 FORMAT(1H1,10X,4A4,' IN LAYER',I3,' AT END OF TIME STEP',I3,
C 1 ' IN STRESS PERIOD',I3/11X,71('-'))
C
C7-----PRINT COLUMN NUMBERS ABOVE THE STRIP
C CALL UCOLNO(J1,J2,ISP,NCAP,NCPF,IOUT)
C
C8-----LOOP THROUGH THE ROWS PRINTING COLS J1 THRU J2 WITH FORMAT IP
C DO 1000 I=1,NROW
C GO TO(10,20,30,40,50,60,70,80,90,100,110,120), IP
C
C-----FORMAT 10G10.3
C 10 WRITE(IOUT,11) I,(BUF(J,I),J=J1,J2)
C 11 FORMAT(1H0,I3,2X,1PG10.3,10(1X,G10.3))
C GO TO 1000
C
C-----FORMAT 8G13.6

```

```

20 WRITE(IOUT,21) I,(BUF(J,I),J=J1,J2)
21 FORMAT(1H0,I3,2X,1PG13.6,8(1X,G13.6))
GO TO 1000
C
C-----FORMAT 15F7.1
30 WRITE(IOUT,31) I,(BUF(J,I),J=J1,J2)
31 FORMAT(1H0,I3,1X,15(1X,F7.1))
GO TO 1000
C
C-----FORMAT 15F7.2
40 WRITE(IOUT,41) I,(BUF(J,I),J=J1,J2)
41 FORMAT(1H0,I3,1X,15(1X,F7.2))
GO TO 1000
C
C-----FORMAT 15F7.3
50 WRITE(IOUT,51) I,(BUF(J,I),J=J1,J2)
51 FORMAT(1H0,I3,1X,15(1X,F7.3))
GO TO 1000
C
C-----FORMAT 15F7.4
60 WRITE(IOUT,61) I,(BUF(J,I),J=J1,J2)
61 FORMAT(1H0,I3,1X,15(1X,F7.4))
GO TO 1000
C
C-----FORMAT 20F5.0
70 WRITE(IOUT,71) I,(BUF(J,I),J=J1,J2)
71 FORMAT(1H0,I3,1X,20(1X,F5.0))
GO TO 1000
C
C-----FORMAT 20F5.1
80 WRITE(IOUT,81) I,(BUF(J,I),J=J1,J2)
81 FORMAT(1H0,I3,1X,20(1X,F5.1))
GO TO 1000
C
C-----FORMAT 20F5.2
90 WRITE(IOUT,91) I,(BUF(J,I),J=J1,J2)
91 FORMAT(1H0,I3,1X,20(1X,F5.2))
GO TO 1000
C
C-----FORMAT 20F5.3
100 WRITE(IOUT,101) I,(BUF(J,I),J=J1,J2)
101 FORMAT(1H0,I3,1X,20(1X,F5.3))
GO TO 1000
C
C-----FORMAT 20F5.4
110 WRITE(IOUT,111) I,(BUF(J,I),J=J1,J2)
111 FORMAT(1H0,I3,1X,20(1X,F5.4))
GO TO 1000
C
C-----FORMAT 9G11.4
120 WRITE(IOUT,121) I,(BUF(J,I),J=J1,J2)
121 FORMAT(1H0,I3,2X,1PG11.4,9(1X,G11.4))
C
1000 CONTINUE
2000 CONTINUE
C
C9-----RETURN
RETURN
END

```

List of Variables for Module ULAPRS

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
BUF	Module	Buffer containing data to be printed or recorded.
I	Module	Index for rows.
ILAY	Module	Layer number.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IP	Module	Format code.
IPRN	Module	Code for the format to be used when printing arrays.
ISP	Module	Number of spaces.
J	Module	Index for columns.
J1	Module	First column in a strip.
J2	Module	Last column in a strip.
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
N	Module	Index for strips.
NCAP	Module	Number of columns on a line.
NCOL	Global	Number of columns in the grid.
NCPF	Module	Number of columns per field.
NROW	Global	Number of rows in the grid.
NSTRIP	Module	Number of strips.
TEXT	Module	Label to be printed or recorded with the array data.


Narrative for Module ULAPRW

Module ULAPRW prints a two-dimensional array in wrap form (fig. 56) using one of twelve FORTRAN formats. Module ULAPRW performs its tasks in the following order:

1. Print a header.
2. Set the format code (IP). If it is less than 1 or greater than 12, set it equal to 12 (the default).
3. Call the module UCOLNO to print column numbers.
4. Loop through the rows printing each one in its entirety using the appropriate format code.
5. RETURN.

Flow Chart for Module ULAPRW

IPRN is a code indicating the format to be used in printing array values. If it is not between 1 and 12, it is set equal to 12.



```

SUBROUTINE ULAPRW(BUF,TEXT,KSTP,KPER,NCOL,NROW,ILAY,IPRN,IOUT)
C
C
C-----VERSION 1642 12MAY1987 ULAPRW
C *****
C PRINT 1 LAYER ARRAY
C *****
C
C SPECIFICATIONS:
C -----
C CHARACTER*4 TEXT
C DIMENSION BUF(NCOL,NROW),TEXT(4)
C -----
C
C1-----PRINT A HEADER
      IF(ILAY.LE.0) GO TO 5
      WRITE(IOUT,1) TEXT,ILAY,KSTP,KPER
      1 FORMAT(1H1,10X,4A4,' IN LAYER',I3,' AT END OF TIME STEP',I3,
      1 ' IN STRESS PERIOD',I3/11X,71('-'))
C
C2-----MAKE SURE THE FORMAT CODE (IP OR IPRN) IS
C2-----BETWEEN 1 AND 12.
      5 IP=IPRN
      IF(IP.LT.1 .OR. IP.GT.12) IP=12
C
C3-----CALL THE UTILITY MODULE UCOLNO TO PRINT COLUMN NUMBERS.
      IF(IP.EQ.1) CALL UCOLNO(1,NCOL,0,11,11,IOUT)
      IF(IP.EQ.2) CALL UCOLNO(1,NCOL,0,9,14,IOUT)
      IF(IP.GT.2 .AND. IP.LT.7) CALL UCOLNO(1,NCOL,3,15,8,IOUT)
      IF(IP.GT.6 .AND. IP.LT.12) CALL UCOLNO(1,NCOL,3,20,6,IOUT)
      IF(IP.EQ.12) CALL UCOLNO(1,NCOL,0,10,12,IOUT)
C
C4-----LOOP THROUGH THE ROWS PRINTING EACH ONE IN ITS ENTIRETY.
      DO 1000 I=1,NROW
      GO TO(10,20,30,40,50,60,70,80,90,100,110,120), IP
C
C-----FORMAT 11G10.3
      10 WRITE(IOUT,11) I,(BUF(J,I),J=1,NCOL)
      11 FORMAT(1H0,I3,2X,1PG10.3,10(1X,G10.3)/(5X,11(1X,G10.3)))
      GO TO 1000
C
C-----FORMAT 9G13.6
      20 WRITE(IOUT,21) I,(BUF(J,I),J=1,NCOL)
      21 FORMAT(1H0,I3,2X,1PG13.6,8(1X,G13.6)/(5X,9(1X,G13.6)))
      GO TO 1000
C
C-----FORMAT 15F7.1
      30 WRITE(IOUT,31) I,(BUF(J,I),J=1,NCOL)
      31 FORMAT(1H0,I3,1X,15(1X,F7.1)/(5X,15(1X,F7.1)))
      GO TO 1000

```

```

C
C----- FORMAT 15F7.2
40 WRITE(IOUT,41) I,(BUF(J,I),J=1,NCOL)
41 FORMAT(1H0,I3,1X,15(1X,F7.2)/(5X,15(1X,F7.2)))
GO TO 1000

C
C----- FORMAT 15F7.3
50 WRITE(IOUT,51) I,(BUF(J,I),J=1,NCOL)
51 FORMAT(1H0,I3,1X,15(1X,F7.3)/(5X,15(1X,F7.3)))
GO TO 1000

C
C----- FORMAT 15F7.4
60 WRITE(IOUT,61) I,(BUF(J,I),J=1,NCOL)
61 FORMAT(1H0,I3,1X,15(1X,F7.4)/(5X,15(1X,F7.4)))
GO TO 1000

C
C----- FORMAT 20F5.0
70 WRITE(IOUT,71) I,(BUF(J,I),J=1,NCOL)
71 FORMAT(1H0,I3,1X,20(1X,F5.0)/(5X,20(1X,F5.0)))
GO TO 1000

C
C----- FORMAT 20F5.1
80 WRITE(IOUT,81) I,(BUF(J,I),J=1,NCOL)
81 FORMAT(1H0,I3,1X,20(1X,F5.1)/(5X,20(1X,F5.1)))
GO TO 1000

C
C----- FORMAT 20F5.2
90 WRITE(IOUT,91) I,(BUF(J,I),J=1,NCOL)
91 FORMAT(1H0,I3,1X,20(1X,F5.2)/(5X,20(1X,F5.2)))
GO TO 1000

C
C----- FORMAT 20F5.3
100 WRITE(IOUT,101) I,(BUF(J,I),J=1,NCOL)
101 FORMAT(1H0,I3,1X,20(1X,F5.3)/(5X,20(1X,F5.3)))
GO TO 1000

C
C----- FORMAT 20F5.4
110 WRITE(IOUT,111) I,(BUF(J,I),J=1,NCOL)
111 FORMAT(1H0,I3,1X,20(1X,F5.4)/(5X,20(1X,F5.4)))
GO TO 1000

C
C----- FORMAT 10G11.4
120 WRITE(IOUT,121) I,(BUF(J,I),J=1,NCOL)
121 FORMAT(1H0,I3,2X,1PG11.4,9(1X,G11.4)/(5X,10(1X,G11.4)))

C
1000 CONTINUE
C
C5-----RETURN
RETURN
END

```

List of Variables for Module ULAPRW

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
BUF	Module	Buffer containing data to be printed or recorded.
I	Module	Index for rows.
ILAY	Module	Layer number.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IP	Module	Format code.
IPRN	Module	Code for the format to be used when printing arrays.
J	Module	Index for columns.
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
NCOL	Global	Number of columns in the grid.
NROW	Global	Number of rows in the grid.
TEXT	Module	Label to be printed or recorded with the array data.